

Proc Tabulate®: A Tool for Evaluating Performance

Lisa P. Wayte
University of Central Florida

Introduction

In a University environment, the College Administrators need to know how well their faculty or staff member is teaching their students. One way to determine this is to create a survey, usually on scantron. Next, the students evaluate the course after several weeks in attendance. Then the information scanned in and analyzed. This type of tool can give Administrators information such as: Are the course materials useful; How can we improve the content of a class; And how well does this class prepare the student for real world applications.

The analysis tool that is used is Proc Tabulate®. Proc Tabulate® performs a summary in a table format of the students' choices for the class and instructor. This information is summarized for a particular class, a department, and a college of the University.

Input Data

The data for the analysis is read in by a NCS scantron scanner using a NCS scantron survey forms. The student chooses their response for each question by filling in the appropriate circle in each category. After the surveys are collected by each department, they are given to the Test Scoring Center for processing. See Appendix A for a sample survey form.

After the surveys are scanned in, an ASCII file is produced. The data is specifically formatted in this manner, a header record followed by the student responses for each question. Since the file contains a lot of classes, there is one header record separating each class. All classes for a particular department are in this file. See Appendix B for sample data file.

Processing the Data

The ASCII data file produced by the NCS scantron is read in by a SAS® data step producing two data sets. One data set contains the header record and the other data set contains the student responses. (If a student response is invalid, it is set to missing.) Now, the two data sets are sorted by class and then combined together.

Next, reshape the answers read in from the data file for each questioned asked. After the data is reshaped, Proc Format® is used to format the data and output with meaningful response categories and questions.

Finally, Proc Tabulate® is used to produce the tabulated results for each individual class, department, and college if needed. Below is a sample SAS® program.

```
options nodate nonumber fmtsearch=(reports);
libname reports '/home/saspgms';
filename sample '/home/data/ascmp.data';
libname ssd01 '/home/sasdata';

data header (keep=dept college inst classes
              lineinfo)
      response (keep=classes ans1-ans16);
  retain course sect classes;

INFILE sample;

length type $ 1;
length dept $ 34;
length college $ 11;
length inst $ 15;
length course $ 8;
length sect $ 4;
length classes $ 12;
length lineinfo $ 60;

INPUT
  @1 type $ @;

select;
  when (type in
        ('A','B','C','E','H','I','J','M'))
  do;
    input @1 college $ 1-15 dept $ 17-34 inst $
           course $ sect $;
    classes = course||sect;
    lineinfo='CLASS:'||classes||' '||'INST:'
            ||inst;
    output header;
  end;
```

```

otherwise
do;
input @1 ans1 1. ans2 1. ans3 1. ans4 1.
ans5 1. ans6 1. ans7 1. ans8 1. ans9 1.
ans10 1. ans11 1. ans12 1. ans13 1.
ans14 1. ans15 1. ans16 1.;

if ans1 > 5 then ans1 = .;
if ans2 > 5 then ans2 = .;
if ans3 > 5 then ans3 = .;
if ans4 > 5 then ans4 = .;
if ans5 > 5 then ans5 = .;
if ans6 > 5 then ans6 = .;
if ans7 > 5 then ans7 = .;
if ans8 > 5 then ans8 = .;
if ans9 > 5 then ans9 = .;
if ans10 > 5 then ans10 = .;
if ans11 > 5 then ans11 = .;
if ans12 > 5 then ans12 = .;
if ans13 > 5 then ans13 = .;
if ans14 > 5 then ans14 = .;
if ans15 > 5 then ans15 = .;
if ans16 > 5 then ans16 = .;
output response;
end;
end;
run;

proc sort data=header; by classes;
run;

proc sort data=response; by classes;
run;

Data combined;
merge header response; by classes;
proc sort data=combined;
by college dept classes inst lineinfo;
run;

data ssd01.ascmp; set combined;
score=ans1; question=1; output;
score=ans2; question=2; output;
score=ans3; question=3; output;
score=ans4; question=4; output;
score=ans5; question=5; output;
score=ans6; question=6; output;
score=ans7; question=7; output;
score=ans8; question=8; output;
score=ans9; question=9; output;
score=ans10; question=10; output;
score=ans11; question=11; output;
score=ans12; question=12; output;
score=ans13; question=13; output;
score=ans14; question=14; output;
score=ans15; question=15; output;
score=ans16; question=16; output;
keep score question classes college dept inst
lineinfo;

proc format library=reports;
value respfmt 1='Excellent'
2='Very Good'
3='Good'
4='Fair'
5='Poor';

value questfmt
1='Feedback to student for performance in course'
2='Instructor encouraged questions from students'
3='Use of class time'
4='Instructor overall organization of course'
5='Continuity between lectures'
6='Pace of course'
7='Instructor assessment of progress used for
class'
8='Text and related materials used'
9='Objectives clearly stated & met'
10='Communication of ideas & info'
11='Express expectations for performance'
12='Available to assist students'
13='Rating of facility & equipment'
14='Instructor knowledgeable of topics'
15='Instructor stimulated interest'
16='Overall assessment of instructor';
run;

/*---Produce individual class summaries---*/

```

```

proc tabulate data=ssd01.ascmp;
class lineinfo question score;
format score respfmt. question questfmt.;
keylabel n='Number Responding';
table lineinfo, question, (score=' '*
(pctn<score all>='% Response by Cat.') n)
/ misstext = '0' printmiss;

Title 'University of Central Florida';
Title2 'Student Perception of Instruction';
Title3 'Spring 1996 Semester: Class Summary
Report';
Title4 'Computer Science';
run;

/*-----Produce departmental summaries-----*/

proc tabulate data=ssd01.ascmp;
class question score;
table question, (score=' '*
(pctn<score all>='% Response by Cat.') n)
/ misstext = '0' printmiss condense;

format score respfmt. question questfmt.;
keylabel n='Number Responding';

Title 'University of Central Florida';
Title2 'Student Perception of Instruction';
Title3 'Spring 1996 Semester: Departmental
Summary Report';
Title4 'Computer Science';
run;

```

Results

The results from Proc Tabulate[®] are produced in a tabular form. Each table lists the course code and section, name of the instructor, the questions asked on the survey, the categories for responses, and the total number of respondents for each question. The responses to the categories are displayed in percentages. If there are no responses for a given category then those responses are indicated by a zero in the category. See Appendix C for sample results.

Conclusions

The program was developed to replace an outdated existing COBOL[®] program. With SAS's flexibility and maintainability, future programs and surveys can be easily updated. This type of data collection and analysis could be used for any type of class or seminar.

Contact Information: Lisa Wayte, University of Central Florida, Computer Services, P. O. Box 162500, Orlando, FL 32816-2500. **Email:** Lisa@mail.ucf.edu **Phone:** (407) 823-5067

SAS is a registered trademark or trademarks of SAS Institute Inc. in the USA and other countries. © indicates USA registration.

Appendix A -- Sample Survey Form

Below is the sample survey form that is used for the analysis. Each student in a class received one NCS scantron survey. The student fills out the following areas: subject, instructor's name, and course prefix and section number. The student marks their response in the circles to the right of each question. Each survey form is treated as an observation in the data file for that particular class.

 Subject

 Instructor

 Course Prefix and Section Number

For items 1-16, please rate the instruction on each of these items concerning the conduct of class.

(If an item appears inappropriate for this course, please leave it blank.)

		Excellent	Very Good	Good	Fair	Poor
1. Feedback concerning your performance in this course was:	E	VG	G	F	P	
2. The instructor encouraged questions from students:	E	VG	G	F	P	
3. Use of class time was:	E	VG	G	F	P	
4. The instructor's overall organization of the course was:	E	VG	G	F	P	
5. Continuity from one class meeting to the next was:	E	VG	G	F	P	
6. The pace of the course was:	E	VG	G	F	P	
7. The instructor's assessment of your progress in the course was:	E	VG	G	F	P	
8. The texts and supplemental learning materials used in the course was:	E	VG	G	F	P	
9. Objectives clearly stated and met:	E	VG	G	F	P	
10. Communication of ideas and information:	E	VG	G	F	P	
11. Expression of expectations for performance:	E	VG	G	F	P	
12. Availability to assist students in or outside of class:	E	VG	G	F	P	
13. Rating of this facility and equipment:	E	VG	G	F	P	
14. Instructor knowledgeable of topics:	E	VG	G	F	P	
15. Instructor stimulated interest:	E	VG	G	F	P	
16. Overall assessment of instructor:	E	VG	G	F	P	

Appendix B -- Sample Data File

The sample data file produced by reading the NCS scantron survey form is shown below. Each class is identified with a header record followed by its data. In the header record, it contains the college name, department name, instructor, course code and section. As you can see, there are many classes that belong to this department. Each department that is requesting the analysis will have a similar file produced. This data file is used by the SAS program for analysis.

Please note: 1) The header record is a variable length line because on the instructor's name. 2) The response data is a fixed format line.

```

Arts & Sciences Computer Science Catbert COP3230 0001
2324333343322432.....
2111111111111111.....
1111111111111111.....
111111111211111.....
1122221421111221.....
Arts & Sciences Computer Science Ratbert COP2131 0002
111111222121121.....
1111111111111111.....
2321113322211231.....
1111111111111111.....
2122132312111221.....
223332234322332.....
1121121311221121.....
222223323332222.....
232222223232323.....
Arts & Sciences Computer Science Dogbert COP2131 0001
3222332332332332.....
1111223221311222.....
333223332312222.....
222133233222222.....
1111121211111111.....
1111111111111111.....
3221232212232222.....
213211223.332132.....
Arts & Sciences Mathematics Dilbert MAN4210 0001
3322122312213332.....
1221122322211232.....
1111111111111111.....
2211123333232333.....
2221222322211222.....

```

Appendix C -- Sample Results

University of Central Florida
 Student Perception of Instruction
 Spring 1996 Semester: Class Summary Report
 Computer Science

CLASS: COP2131 0001 INST: DOGBERT

QUESTION	Excellent	Very Good	Good	Fair	Poor	Number Responding
	% Response by Cat.	% Response by Cat.	% Response by Cat.	% Response by Cat.	% Response by Cat.	
Feedback to student for performance in course	37.50	25.00	37.50	0	0	8.00
Instructor encouraged questions from students	50.00	37.50	12.50	0	0	8.00
Use of class time	37.50	37.50	25.00	0	0	8.00
Instructor overall organization of course	62.50	37.50	0	0	0	8.00
Continuity between lectures	37.50	37.50	25.00	0	0	8.00
Pace of course	25.00	25.00	50.00	0	0	8.00
Instructor assessment of progress used for class	25.00	50.00	25.00	0	0	8.00
Text and related materials used	12.50	50.00	37.50	0	0	8.00
Objectives clearly stated & met	37.50	25.00	37.50	0	0	8.00
Communication of ideas & info	42.86	42.86	14.29	0	0	7.00
Express expectations for performance	37.50	25.00	37.50	0	0	8.00
Available to assist students	37.50	25.00	37.50	0	0	8.00
Rating of facility & equipment	37.50	62.50	0	0	0	8.00
Instructor knowledge of topics	37.50	50.00	12.50	0	0	8.00
Instructor stimulated interest	25.00	50.00	25.00	0	0	8.00
Overall assessment of instructor	25.00	75.00	0	0	0	8.00

University of Central Florida
 Student Perception of Instruction
 Spring 1996 Semester: Departmental Summary Report
 Computer Science

QUESTION	Excellent	Very Good	Good	Fair	Poor	Number Responding
	% Response by Cat.	% Response by Cat.	% Response by Cat.	% Response by Cat.	% Response by Cat.	
Feedback to student for performance in course	43.55	40.32	12.90	3.23	0	62.00
Instructor encouraged questions from students	55.56	26.98	17.46	0	0	63.00
Use of class time	46.03	39.68	14.29	0	0	63.00
Instructor overall organization of course	63.49	30.16	3.17	3.17	0	63.00
Continuity between lectures	53.23	35.48	9.68	1.61	0	62.00
Pace of course	31.75	30.16	31.75	6.35	0	63.00
Instructor assessment of progress used for class	42.86	34.92	20.63	1.59	0	63.00
Text and related materials used	26.98	26.98	38.10	7.94	0	63.00
Objectives clearly stated & met	42.86	34.92	17.46	4.76	0	63.00
Communication of ideas & info	41.94	41.94	12.90	3.23	0	62.00
Express expectations for performance	44.44	34.92	20.63	0	0	63.00
Available to assist students	58.73	25.40	15.87	0	0	63.00
Rating of facility & equipment	63.49	25.40	9.52	1.59	0	63.00
Instructor knowledge of topics	39.68	38.10	17.46	3.17	1.59	63.00
Stimulation of interest	31.75	42.86	23.81	1.59	0	63.00
Overall assessment of instructor	66.67	25.40	6.35	1.59	0	63.00